Confidence Intervals for Binomial Proportion Using SAS®: The All You Need to Know and No More…
Jiangtang Hu
d-Wise, Morrisville, NC

ABSTRACT
Confidence Intervals (CI) are extremely important in presenting clinical results. The choosing of right algorithms of CI is the plate of statisticians, and this paper is for SAS programmers where more than 14 methods to compute CI for single proportion is presented with executable SAS codes, by SAS procedures and customized codes from the scratch.

These codes is currently hosted in my Github page:
https://raw.githubusercontent.com/Jiangtang/Programming-SAS/master/CI_Single_Proportion.sas

Some commentaries from A SAS programmer’s point of view will also be presented.

INTRODUCTION
Suppose n is the same size, r the number of count of interested outcome, and \( p = \frac{r}{n} \) is so called binomial proportion (sample proportion). A confidence interval (CI) is a range of values, computed from the sample, which is with probability of 95% to cover the population proportion (well, you may use any pre-specified probabilities, but 95% is the most common one). From statistical point of view, confidence intervals are generally more informative than p-value. In clinical studies, the size of difference of the outcome between groups (measured by confidence intervals) is much more useful for researchers than the single significant indicator, namely, p-value [1].

For SAS programmers, it’s nice to know that the confidence interval is much preferred to p-value for presenting clinical outcomes. Computing varieties of confidence intervals (there are A LOT!) is part of SAS programmers’ daily job. The selection of the right type of CI is decided by statisticians, and this paper is primary for SAS programmers: you might not be very familiar with the story behind the different CIs, but with this paper and SAS codes attached, you can always get the right results given the type of CIs is articulated in the Statistical Analysis Plan(SAP).

I will leave the calculation of confidence interval for the difference between independent proportions [2] to another paper and this one is only for single binomial proportion. Particularly, 14 methods will be presented with fully executable SAS programs:

1. Simple asymptotic, without Continuity Correction (CC), mostly know as **Wald**
2. Simple asymptotic, with CC
3. Score method, without CC, also known as **Wilson**
4. Score method, with CC
5. Binomial-based, 'Exact' or **Clopper-Pearson**
6. Binomial-based, Mid-p
7. Likelihood-based
8. Jeffreys
9. Agresti-Coull, pseudo frequency, \( z^2/2 \) successes| psi = \( z^2/2 \)
10. Agresti-Coull, pseudo frequency, 2 successes and 2 failures| psi = 2
11. Agresti-Coull, pseudo frequency, psi = 1
12. Agresti-Coull, pseudo frequency, psi = 3
13. Logit
14. Blaker
To get a quick impression, just run the following piece of code:

```sas
filename CI url 'https://raw.githubusercontent.com/Jiangtang/Programming-SAS/master/CI_Single_Proportion.sas';
%include CI;
%CI_Single_Proportion(r=81, n=263);
```

The output (note: CC is short for Continuity Correction):

Method #1-7 was well documented in a famous paper by Robert Newcombe, *Two-sided confidence intervals for the single proportion: comparison of seven methods* [3] where the corresponding output is:

A. BASIC ALGORITHMS AND COMMENTS

A1 Method #1 and #2

The most popular method in introductory textbooks (note not in practice) is method #1 ('Wald') due to its simplest form:

\[ p \pm z \times SE \]

where
p is the empirical estimate of the proportion, \( r / n \),

SE, standard error = \( \sqrt{(p(1-p)/n)} \)

\( z \), the quantile of the standard normal distribution (1.960 for the usual two-sided 95% interval).

Despite its popularity, the Wald method is very deficient. For example, it is not boundary-respecting and it can extend beyond 0 or 1.

When \( p = 0 \) or \( 1 \), method #1 (‘Wald’) will get a zero width interval [0, 0]. To avoid this degeneracy issue, method #2 (‘Wald with CC’) introduces a continuity correction (CC), namely, \( 1/(2n) \). But be aware of the trade-off, the problem of adding the continuity correction is that it would lead to more instances of overshoot. A step-by-step approach to demonstrate these two methods is as follows:

```
data m1;
r = 81;
n = 263;
alpha = 0.05;
p = r / n;
z = probit (1-alpha/2);
*standard error;
se = (sqrt(n*p*(1-p)))/n;
L = p - z * se;
U  = p + z * se;
put L= U= ;
run;
```

```
data m2;
r = 81;
n = 263;
alpha = 0.05;
p = r / n;
z = probit (1-alpha/2);
*standard error;
se = (sqrt(n*p*(1-p)))/n;
*continuity correction;
cc = 1/(2*n);
L = p - (z * se + cc);
U = p + (z * se + cc);
put L= U=;
run;
```

**A2 Method #13**

The Logit method (#13) is actually very similar to Wald method in structure. Instead using \( p = r / n \) as the empirical proportion, a logit transformation is applied in Logit method, namely, \( \exp[\log(p/(1-p)))] \):

```
data m13;
r = 81;
n = 263;
alpha = 0.05;
p = r / n;
z = probit (1-alpha/2);
L = exp(log(p/(1-p)) - z*sqrt(n/(r*(n-r)))) / (1+exp(log(p/(1-p)) - z*sqrt(n/(r*(n-r)))));
U = exp(log(p/(1-p)) + z*sqrt(n/(r*(n-r)))) / (1+exp(log(p/(1-p)) + z*sqrt(n/(r*(n-r)))));
put L= U=;
run;
```

Logit method is often used for odds ratios. Like the Wald method, it is not guaranteed satisfactory when \( n \) is small or \( p \) is close to 0 or 1.

**A3 Method #3, #4**

Wilson score method (#3) is considered the simplest acceptable alternative to the Wald approach. It gets better performance when \( n \) is small and when \( p \) is close to 0 or 1.
Wilson method is also not boundary-respecting. A continuity correction can be applied to get method #4:

```sas
data m3;
  r = 81;
  n = 263;
  alpha = 0.05;
  p = r / n;
  q = 1-p;
  z = probit (1-alpha/2);
  L = ( 2*r+z**2 - (z*sqrt(z**2+4*r*q)) ) / (2*(n+z**2));
  U = ( 2*r+z**2 + (z*sqrt(z**2+4*r*q)) ) / (2*(n+z**2));
  put L= U=;
run;
```

A4 Method #5

The Wald-like intervals described as above are all asymptotic intervals. The so called Clopper-Pearson 'exact' method (#5) is quite different since it's very conservative. It’s very computationally convenient and only one inverse Beta function is used:

```sas
data m5;
  r = 81;
  n = 263;
  alpha = 0.05;
  L = 1 - betainv(1 - alpha/2,n-r+1,r);
  U = betainv(1 - alpha/2,r+1 ,n-r);
  put L= U=;
run;
```

A5 Method #6

The method #5 is so conservative that sometimes it's even unnecessary. A similar 'mid-p' (#6) is used to reduce the conservatism. Method #6 accumulates the tail areas and it's relatively complicated compared to any methods mentioned above. The full code is available in the master code (of session 'method 6'):

https://raw.github.com/Jiangtang/Programming-SAS/master/CI_Single_Proportion.sas

A6 Method #7

Likelihood-based approach (#7) is said to be theoretically appealing. The full program is available here (of session 'method 7'):

https://raw.github.com/Jiangtang/Programming-SAS/master/CI_Single_Proportion.sas

A7 Method #9-#12

Methods #9-#12 come from the family of 'pseduo-frequency methods'. Instead of using p = r / n, 4 methods are developed when \( \psi \) (pronounced as psi) gets values \( \frac{z}{2}, 2, 1, 3 \):

\[
p_{\psi} = \frac{(r + \psi)}{(n + 2\psi)}
\]

```sas
data m9_12;
```

```sas

data m4;
  r = 81;
  n = 263;
  alpha = 0.05;
  p = r / n;
  q = 1-p;
  z = probit (1-alpha/2);
  L = ( 2*r+z**2 -1 - z*sqrt(z**2 - 2- 1/n + 4*p*(n*q1))) / (2*(n+z**2));
  U = ( 2*r+z**2 +1 + z*sqrt(z**2 + 2- 1/n + 4*p*(n*q-1))) / (2*(n+z**2));
  put L= U=;
run;
```
A8 Method #8

The methods mentioned above are all frequentist methods. The alternative Bayesian methods are sometimes preferable because they incorporate prior information. I'm not well equipped with Bayesian methods, but one called 'Jeffreys' method is pretty straightforward from programming point of view:

```sas
data m8;
  r = 81;
  n = 263;
  alpha = 0.05;
  L = betainv( alpha/2, r+0.5, n-r+0.5);
  U = betainv(1-alpha/2, r+0.5, n-r+0.5);
  put L= U=;
run;
```

A9 Method #14

Although CIs using the Blaker method are available with options from a SAS FREQ procedure, the program below manually computes the CI lower and upper bounds using the Blaker method:

```sas
data m14;
  r = 81;
  n = 263;
  alpha = 0.05;
  tolerance=1e-05;
  lower = 0;
  upper = 1;
  if r ^= 0 then do;
    lower = quantile('BETA',alpha/2, r, n-r+1);
    do while (acceptbin(r, n, lower + tolerance) < (alpha));
      lower = lower + tolerance;
    end;
  end;
  if r ^= n then do;
    upper = quantile('BETA', 1 - alpha/2, r+1, n-r);
    do while (acceptbin(r, n, upper - tolerance) < (alpha));
      upper = upper - tolerance;
    end;
end;
```
B. THE WORKHORSE: SAS FREQ

The individual SAS programs above can be used for educational purpose to better understand the various methods one can use to calculate CIs. I compiled them together into a macro and can be called as

```sas
filename CI url 'https://raw.githubusercontent.com/Jiangtang/Programming-SAS/master/CI_Single_Proportion.sas';
%include CI;
%CI_Single_Proportion(r=81,n=263);
```

which can print out all outputs from the 14 methods together for comparison.

Alternatively, SAS PROC FREQ (I use SAS 9.4 TS1M2 in Windows 7 when writing this paper) can be used to compute most of the confidence intervals (11 to be precisely):

```sas
data test;
input grp outcome $ count;
datalines;
1 f 81
1 u 182;
/*2.3 CL=enumeration: #1,#3, #5,#6, #7,#8, #9,#13, #14*/
ods select BinomialCLs;
proc freq data=test;
   tables outcome / binomial (CL=
    WALD
    WILSON
    CLOPPERPEARSON
    MIDP
    LIKELIHOODRATIO
    JEFFREYS
    AGRESTICOULL
    LOGIT
    BLAKER
    );
   weight Count;
run;
/*2.4 CL=CC : #2,#4*/
ods select BinomialCLs;
proc freq data=test;
   tables outcome / binomial (CL=
    WILSON(CORRECT)
    WALD(CORRECT)
    );
   weight Count;
run;
```

The output:
Only methods #10 - #12 from ‘pseduo-frequency methods’ family are not available in SAS PROC FREQ.

A programming note:

1) If no confidence interval option assigned, the Wald and the ‘exact’ CIs will be presented.
2) If option CL = All is applied, the following 5 CIs will be computed: Agresti-Coull, Clopper-Pearson (Exact), Jeffreys, Wald, Wilson.

C. OTHER SAS PRECEDURES

SAS PROC MCMC can be used to compute Bayesian confidence interval but I’m not studying it yet.

D. CONCLUSION

SAS PROC FREQ is pretty impressive to compute varieties of confidence intervals. The customized SAS programs in this paper are best used for educational purposes, or to serve as a start to customize your own pieces of CI programs.
REFERENCES


ACKNOWLEDGMENTS

I highly recommend Robert Newcombe’s book and papers to jump into the world of confidence intervals. I personally gain a lot from them.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Jiangtang (JT) Hu
Life Sciences Consultant
d-Wise Technologies, Inc.
1500 Perimeter Park Dr., Suite 150, Morrisville, NC, 27560
www.d-Wise.com

(O) 919-334-6096
(C) 919-801-9659
(F): 888-563-0931
(O) Jiangtang_Hu@d-Wise.com
(P) jiangtanghu@gmail.com
http://jiangtanghu.com/

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.